Appendix A
Linear Model Equations and Mplus Syntax:
MAR Model (Figure 1)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (1)$$

$$\eta_{0i} = \alpha_0 + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (2)$$

$$\eta_{1i} = \alpha_1 + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (3)$$

$$\eta_{2i} = \alpha_2 \quad (4)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12;

Missing = ALL (-99);

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0; !Non-significant quadratic variance fixed at zero;

I S ON RT SFAS;

OUTPUT: Patterns;
!Will produce the 8 missing data patterns shown in Table 1;
```

Appendix A1
Jump to Reference
Step 1: Imputation

```
DATA:
FILE IS JTR.dat;
!This file excludes treatment participants with complete data;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12;

Missing = ALL (-99);

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0; !Non-significant quadratic variance fixed at zero;

I S ON RT SFAS;

DATA IMPUTATION:
NDATASETS = 20;
IMPUTE = BINGE_1 BINGE_6 BINGE_12;
SAVE = BINGEimp*.dat;
```

Appendix A2
Jump to Reference
Step 2: Analysis

```
DATA:
FILE IS MergeImplist.dat;
TYPE = IMPUTATION;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0; !Non-significant quadratic variance fixed at zero;

I S ON RT SFAS;
```

Appendix A3
Delta Method
Step 1: Imputation

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12;

Missing = ALL (-99);

AUXILIARY = ID; !ID will be included in all imputations;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0; !Non-significant quadratic variance fixed at zero;

I S ON RT SFAS;

DATA IMPUTATION:
NDATASETS = 20;
IMPUTE = BINGE_1 BINGE_6 BINGE_12;
SAVE = DELTA_imp*.dat;
```

Appendix A4
Delta Method
Step 2: Analysis

```
DATA:
FILE IS MergeImplist.dat;
TYPE = IMPUTATION;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12;

DEFINE:
IF (RT EQ 1 AND PATTERN EQ 2 OR PATTERN EQ 6)
THEN BINGE_12 = BINGE_12 + 1.04;
IF (SFAS EQ 1 AND PATTERN EQ 2 OR PATTERN EQ 6)
THEN BINGE_12 = BINGE_12 + 1.04;

IF (RT EQ 1 AND PATTERN EQ 4)
THEN BINGE_6 = BINGE_6 + 1.04;
IF (SFAS EQ 1 AND PATTERN EQ 4)
THEN BINGE_6 = BINGE_6 + 1.04;

IF (RT EQ 1 AND PATTERN EQ 4)
THEN BINGE_12 = BINGE_6 + 1.04;
IF (SFAS EQ 1 AND PATTERN EQ 4)
THEN BINGE_12 = BINGE_6 + 1.04;

IF (RT EQ 1 AND PATTERN EQ 8)
THEN BINGE_1 = BINGE_1 + 1.04;
IF (SFAS EQ 1 AND PATTERN EQ 8)
THEN BINGE_1 = BINGE_1 + 1.04;

IF (RT EQ 1 AND PATTERN EQ 8)
THEN BINGE_6 = BINGE_6 + 1.04;
IF (SFAS EQ 1 AND PATTERN EQ 8)
THEN BINGE_6 = BINGE_6 + 1.04;

IF (RT EQ 1 AND PATTERN EQ 8)
THEN BINGE_12 = BINGE_6 + 1.04;
IF (SFAS EQ 1 AND PATTERN EQ 8)
THEN BINGE_12 = BINGE_6 + 1.04;

MODEL:
  I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
  Q@0;
```

Appendix B
Linear Model Equations and Mplus Syntax:
Focal Pattern Mixture Model (Figure 2)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (5)$$

$$\eta_{0i} = \alpha_0 + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \gamma_{03}(d1_i) + \gamma_{03}(d2_i) + \gamma_{03}(d3_i) + \zeta_{0i} \quad (6)$$

$$\eta_{1i} = \alpha_1 + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \gamma_{13}(d1_i) + \gamma_{13}(d2_i) + \gamma_{13}(d3_i) + \zeta_{1i} \quad (7)$$

$$\eta_{2i} = \alpha_2 \quad (8)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 d1-d3;

Missing = ALL (-99);

!This syntax creates the three (d1-d3) dummy missing data codes;
DATA MISSING:
names = BINGE_B BINGE_1 BINGE_6 BINGE_12;
type = ddropout;
binary = d1-d3;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;
I ON d1-d3(2);
!Constrained to equality for model identification;
S ON d1-d3(3);
!Constrained to equality for model identification;
```

Appendix C
Linear Model Equations and Mplus Syntax:
Diffuse Pattern Mixture Model (Figure 3)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \tag{9}$$

$$\eta_{0i} = \alpha_0 + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \gamma_{03}(d1_i) + \gamma_{03}(d2_i) + \gamma_{03}(d3_i) +$$

$$\gamma_{04}(d1_i - d3_i * RT_i) + \gamma_{05}(d1_i - d3_i * SFAS_i) + \zeta_{0i} \tag{10}$$

$$\eta_{1i} = \alpha_1 + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \gamma_{13}(d1_i) + \gamma_{13}(d2_i) + \gamma_{13}(d3_i)$$

$$+\gamma_{14}(d1_i - d3_i * RT_i) + \gamma_{15}(d1_i - d3_i * SFAS_i) + \zeta_{1i} \tag{11}$$

$$\eta_{2i} = \alpha_2 \tag{12}$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12 d1-d3;

USEVARIABLES = RT - BINGE_12 d1-d3 rt_di1 rt_di2 rt_di3 sfas_di1 sfas_di2
sfas_di3;

Missing = ALL (-99);

DEFINE:
rt_di1=RT*d1;
rt_di2=RT*d2;
rt_di3=RT*d3;
sfas_di1=sfas*d1;
sfas_di2=sfas*d2;
sfas_di3=sfas*d3;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;
!All separately constrained to equality for identification;
I ON d1-d3(2);
S ON d1-d3(3);
I ON rt_di1-rt_di3(4);
S ON rt_di1-rt_di3(5);
I ON sfas_di1-sfas_di3(6);
S ON sfas_di1-sfas_di3(7);
```

Appendix D
Linear Model Equations and Mplus Syntax:
Diffuse Shared Parameter Model #1 (Figure #4)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (13)$$

$$\eta_{0i} = \alpha_0 + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (14)$$

$$\eta_{1i} = \alpha_1 + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (15)$$

$$\eta_{2i} = \alpha_2 \quad (16)$$

$$Logit(s1_i) = \pi_{10} + \pi_1(\eta_{0i}) + \pi_2(\eta_{1i}) \quad (17)$$

$$Logit(s2_i) = \pi_{20} + \pi_1(\eta_{0i}) + \pi_2(\eta_{1i}) \quad (18)$$

$$Logit(s3_i) = \pi_{30} + \pi_1(\eta_{0i}) + \pi_2(\eta_{1i}) \quad (19)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 S1-S3;

Missing = ALL (-99);

categorical are S1-S3;
!Survival missing data indicators are binary response variables and must
be listed here;

!This creates the survival (s1-s3) missing data codes;
DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;

!Equality constraints for identification purposes;
S1-S3 on I(4);
S1-S3 on S(5);
```

Appendix E
Linear Model Equations and Mplus Syntax:
Diffuse Shared Parameter Model #2 (Figure #5)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (20)$$

$$\eta_{0i} = \alpha_0 + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (21)$$

$$\eta_{1i} = \alpha_1 + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (22)$$

$$\eta_{2i} = \alpha_2 \quad (23)$$

$$Logit(s1_i) = \pi_{10} + \pi_1(\eta_{0i}) + \pi_2(\eta_{1i}) + \pi_3(RT_i) + \pi_4(SFAS_i) \quad (24)$$

$$Logit(s2_i) = \pi_{20} + \pi_1(\eta_{0i}) + \pi_2(\eta_{1i}) + \pi_3(RT_i) + \pi_4(SFAS_i) \quad (25)$$

$$Logit(s3_i) = \pi_{30} + \pi_1(\eta_{0i}) + \pi_2(\eta_{1i}) + \pi_3(RT_i) + \pi_4(SFAS_i) \quad (26)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 S1-S3;

Missing = ALL (-99);

categorical are S1-S3;
!Binary survival missing data indicators must be listed here;

!This creates the survival (s1-s3) missing data codes;
DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;

!Equality constraints for identification purposes;
S1-S3 on RT(2);
S1-S3 on SFAS(3);
S1-S3 on I(4);
S1-S3 on S(5);
```

Appendix F
Mplus Syntax:
Shared Parameter Model with Multinomial Missing Data Codes

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT – BINGE_12 M1-M3;

Missing = ALL (-99);

!Multinomial codes must be user-created then declared here;
!'CATEGORICAL' can be used for binary or ordinal response variables;
CATEGORICAL are M1-M3;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;

!Equality constraints for identification purposes;
M1-M3 on RT(2);
M1-M3 on SFAS(3);
M1-M3 on I(4);
M1-M3 on S(5);
```

Appendix G
Linear Model Equations and Mplus Syntax:
Yuan & Little (2009) Hybrid Model (Figure #6)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (27)$$

$$\eta_{0i} = \alpha_0 + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + +\gamma_{03}(Droptime_i) + \zeta_{0i} \quad (28)$$

$$\eta_{1i} = \alpha_1 + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \gamma_{13}(Droptime_i) + \zeta_{1i} \quad (29)$$

$$\eta_{2i} = \alpha_2 \quad (30)$$

$$Logit(s1_i) = \pi_{10} + \pi_1(RT_i) + \pi_2(SFAS_i) + \pi_4(\eta_{0i}) + \pi_5(\eta_{1i}) \quad (31)$$

$$Logit(s2_i) = \pi_{20} + \pi_1(RT_i) + \pi_2(SFAS_i) + \pi_4(\eta_{0i}) + \pi_5(\eta_{1i}) \quad (32)$$

$$Logit(s3_i) = \pi_{30} + \pi_1(RT_i) + \pi_2(SFAS_i) + \pi_4(\eta_{0i}) + \pi_5(\eta_{1i}) \quad (33)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12 pattern;

USEVARIABLES = RT - BINGE_12 S1-S3 droptime;

Missing = ALL (-99);

categorical are S1-S3;
!Binary survival missing data indicators must be listed here;

!This creates the survival (s1-s3) missing data codes;
DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

DEFINE: !Creates the 'Droptime' missing data variable, see Table 1;
if (Pattern EQ 1) THEN droptime = 4;
if (Pattern EQ 2) THEN droptime = 3;
if (Pattern EQ 3) THEN droptime = 4;
if (Pattern EQ 4) THEN droptime = 2;
if (Pattern EQ 5) THEN droptime = 4;
If (Pattern EQ 6) THEN droptime = 3;
if (Pattern EQ 7) THEN droptime = 4;
If (Pattern EQ 8) THEN droptime = 1;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
```

```
Q@0;
I S ON RT SFAS Droptime;

!Equality constraints for identification purposes;
S1-S3 on RT(2);
S1-S3 on SFAS(3);
S1-S3 on I(4);
S1-S3 on S(5);
```

Appendix H
Linear Model Equations and Mplus Syntax:
Focal Selection Model (Figure #7)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (34)$$

$$\eta_{0i} = \alpha_0 + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (35)$$

$$\eta_{1i} = \alpha_1 + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (36)$$

$$\eta_{2i} = \alpha_2 \quad (37)$$

$$Logit(s1_i) = \pi_{10} + \pi_{11}(BINGE_{1i}) \quad (38)$$

$$Logit(s2_i) = \pi_{20} + \pi_{11}(BINGE_{2i}) \quad (39)$$

$$Logit(s3_i) = \pi_{30} + \pi_{11}(BINGE_{3i}) \quad (40)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 S1-S3;

Missing = ALL (-99);

categorical are S1-S3;

DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;

!Equality constraints for identification;

s1 ON BINGE_1 (1);
s2 ON BINGE_6 (1);
s3 ON BINGE_12 (1);
```

Appendix I
Linear Model Equations and Mplus Syntax:
Diffuse Selection (Lag Prediction) Model (Figure #8)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (41)$$

$$\eta_{0i} = \alpha_0 + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (42)$$

$$\eta_{1i} = \alpha_1 + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (43)$$

$$\eta_{2i} = \alpha_2 \quad (44)$$

$$Logit(s1_i) = \pi_{10} + \pi_{11}(BINGE_{1i}) + \pi_{12}(BINGE_{t-1i}) \quad (45)$$

$$Logit(s2_i) = \pi_{20} + \pi_{11}(BINGE_{2i}) + \pi_{12}(BINGE_{t-1i}) \quad (46)$$

$$Logit(s3_i) = \pi_{30} + \pi_{11}(BINGE_{3i}) + \pi_{12}(BINGE_{t-1i}) \quad (47)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 S1-S3;

Missing = ALL (-99);

categorical are S1-S3;

DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;

!Equality constraints for identification;

s1 ON BINGE_1 (1);
s2 ON BINGE_6 (1);
s3 ON BINGE_12 (1);
s1 ON BINGE_B (2);
s2 ON BINGE_1 (2);
s3 ON BINGE_6 (2);
```

Appendix J
Linear Model Equations and Mplus Syntax:
Diffuse Selection (IV Group Prediction) Model (Figure #9)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (48)$$

$$\eta_{0i} = \alpha_0 + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (49)$$

$$\eta_{1i} = \alpha_1 + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (50)$$

$$\eta_{2i} = \alpha_2 \quad (51)$$

$$Logit(s1_i) = \pi_{10} + \pi_{11}(BINGE_{1i}) + \pi_{12}(RT_i) + \pi_{13}(SFAS_i) \quad (52)$$

$$Logit(s2_i) = \pi_{20} + \pi_{11}(BINGE_{2i}) + \pi_{12}(RT_i) + \pi_{13}(SFAS_i) \quad (53)$$

$$Logit(s3_i) = \pi_{30} + \pi_{11}(BINGE_{3i}) + \pi_{12}(RT_i) + \pi_{13}(SFAS_i) \quad (54)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 S1-S3;

Missing = ALL (-99);

categorical are S1-S3;

DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;

!Equality constraints for identification;

s1 ON BINGE_1 (1);
s2 ON BINGE_6 (1);
s3 ON BINGE_12 (1);


s1-s3 ON RT (2);
s1-s3 ON SFAS (3);
```

Appendix K
Linear Model Equations and Mplus Syntax:
Diffuse Selection (Lag & IV Group Prediction) Model (Figure #10)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (55)$$

$$\eta_{0i} = \alpha_0 + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (56)$$

$$\eta_{1i} = \alpha_1 + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (57)$$

$$\eta_{2i} = \alpha_2 \quad (58)$$

$$Logit(s1_i) = \pi_{10} + \pi_{11}(BINGE_{1i}) + \pi_{12}(BINGE_{t-1i}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (59)$$

$$Logit(s2_i) = \pi_{20} + \pi_{11}(BINGE_{2i}) + \pi_{12}(BINGE_{t-1i}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (60)$$

$$Logit(s3_i) = \pi_{30} + \pi_{11}(BINGE_{3i}) + \pi_{12}(BINGE_{t-1i}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (61)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 S1-S3;

Missing = ALL (-99);

categorical are S1-S3;

DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;

s1 ON BINGE_1 (1);
s2 ON BINGE_6 (1);
s3 ON BINGE_12 (1);

s1-s3 ON RT (2);
s1-s3 ON SFAS (3);

s1 ON BINGE_B (4); s2 ON BINGE_1 (4);
s3 ON BINGE_6 (4);
```

Appendix L
Mplus Syntax:
Diffuse (Lag & IV Group Prediction) Selection Model with Multinomial Missing Data Codes

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12 M1-M3;

USEVARIABLES = RT - BINGE_12 M1-M3;

Missing = ALL (Your Missing Value Indicator Here);

categorical are m1-m3;

MODEL:
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;

!Equality constraints for identification;
m1-m3 ON RT (1);
m1-m3 ON SFAS (2);

m1 ON BINGE_B (3);
m2 ON BINGE_1 (3);
m3 ON BINGE_6 (3);

m1 ON BINGE_1 (4);
m2 ON BINGE_6 (4);
m3 ON BINGE_12 (4);
```

Appendix M
Linear Model Equations and Mplus Syntax:
MAR Mixture Model (k = 2) (Figure #11)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (62)$$

$$\eta_{0i} = \alpha_{0k} + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (63)$$

$$\eta_{1i} = \alpha_{1k} + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (64)$$

$$\eta_{2i} = \alpha_{2k} \quad (65)$$

$$Logit(C_i|k = 2) = \pi_{00} \quad (66)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12;

Missing = ALL (-99);

CLASSES = c (2);

ANALYSIS:
Type = mixture;

MODEL:
%overall%
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;

%c#1%
[I];
[S];
[Q];

%c#2%
[I];
[S];
[Q];
```

Appendix N
Linear Model Equations and Mplus Syntax:
Roy (2003) Dropout Model ($k = 2$) (Figure #12)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (67)$$

$$\eta_{0i} = \alpha_{0k} + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (68)$$

$$\eta_{1i} = \alpha_{1k} + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (69)$$

$$\eta_{2i} = \alpha_{2k} \quad (70)$$

$$Logit(C_i|k = 2) = \pi_{00} + \pi_{1k}(d1_i) + \pi_{2k}(d2_i) + \pi_{3k}(d3_i) \quad (71)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 d1-d3;

Missing = ALL (-99);

!The number of mixture trajectories to be extracted declared here;
classes = c (2);

DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = ddropout;
binary = d1-d3;

!Needed for mixture model estimation and solution verification;
ANALYSIS:
type = mixture;
ALGORITHM=INTEGRATION;
process = 8(starts);
starts = 400 100;

MODEL:
%OVERALL%
I S Q | binge_b@0 binge_1@.083 binge_6@.5 binge_12@1;
Q@0;

I S ON RT SFAS;

!Dropout effects predict latent mixture trajectories;
c ON d1-d3;
```

Appendix O
Linear Model Equations and Mplus Syntax:
Muthén-Roy Model ($k = 2$, $d = 2$) (Figure #13)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \tag{72}$$

$$\eta_{0i} = \alpha_{0k} + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \tag{73}$$

$$\eta_{1i} = \alpha_{1kd} + \gamma_{11}(RT) + \gamma_{12}(SFAS_i) + \zeta_{1i} \tag{74}$$

$$\eta_{2i} = \alpha_{2kd} \tag{75}$$

$$Logit(C_i|k) = \pi_{00} \tag{76}$$

$$Logit(C_i|d) = \pi_{01} + \pi_{1d}(d1_i) + \pi_{2d}(d2_i) + \pi_{3d}(d3_i) \tag{77}$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 d1-d3;

Missing = ALL (-99);

!ck is the response latent categorical variable;
!cd is the dropout latent categorical variable;
classes = cd(2) ck(2);

Data missing:
names = BINGE_B BINGE_1-BINGE_12;
type = ddropout;
binary = d1-d3;

ANALYSIS:
TYPE = MIXTURE;
ALGORITHM=INTEGRATION;
process = 8(starts);
starts = 400 100;
!Needed for two (cor)related categorical latent variables;
parameterization = loglinear;

MODEL:
%OVERALL%
I S Q | binge_b@0 binge_1@.083 binge_6@.5 binge_12@1;
Q@0;
I S ON RT SFAS;
cd WITH ck;
cd ON d1-d3;
```

```
!Intercept means constrained to equality within ck mixtures;
%cd#1.ck#1%
[I](1);
[S];
[Q];

%cd#1.ck#2%
[I](2);
[S];
[Q];

%cd#2.ck#1%
[I](1);
[S];
[Q];

%cd#2.ck#2%
[I](2);
[S];
[Q];
```

Appendix P
Linear Model Equations and Mplus Syntax:
Beunckens et al. (2008) Model (k = 2) (Figure #14)

$$Y_{ti} = \lambda_1\eta_{0i} + \lambda_t\eta_{1i} + \lambda_{t^2}\eta_{2i} + \varepsilon_{ti} \quad (78)$$

$$\eta_{0i} = \alpha_{0k} + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (79)$$

$$\eta_{1i} = \alpha_{1k} + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (80)$$

$$\eta_{2i} = \alpha_{2k} \quad (81)$$

$$Logit(C_i|k) = \pi_{00} \quad (82)$$

$$Logit(s1_i) = \pi_{10k} + \pi_1(\eta_{0i}) + \pi_2(RT_i) + \pi_3(SFAS_i) \quad (83)$$

$$Logit(s2_i) = \pi^*_{20k} + \pi_1(\eta_{0i}) + \pi_2(RT_i) + \pi_3(SFAS_i) \quad (84)$$

$$Logit(s3_i) = \pi^*_{30k} + \pi_1(\eta_{0i}) + \pi_2(RT_i) + \pi_3(SFAS_i) \quad (85)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT – BINGE_12 s1-s3;

Missing = ALL (-99);

categorical are S1-S3;

classes = c(2);

DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

Analysis:
type = mixture;
ALGORITHM=INTEGRATION;
process = 4(starts);
starts = 400 100;

%OVERALL%
I S Q | binge_b@0 binge_1@.083 binge_6@.5 binge_12@1;
Q@0;
I S ON RT SFAS;
```

```
s1-s3 ON I (1); !constrained to equality;
s1-s3 ON RT (2); !constrained to equality;
s1-s3 ON SFAS (3); !constrained to equality;

!Survival logit intercepts with labels for re-scaling;
%C#1%
 [s1$1] (p2);
 [s2$1] (p3);
 [s3$1] (p4);

%C#2%
 [s1$1] (p12);
 [s2$1] (p13);
 [s3$1] (p14);

MODEL CONSTRAINT:
New(G1 G12 G2 G22);
p3 = p2+g1*.083+g12*.006889; !g1*1st slope loading, g12*1st slope loading2
p4 = p2+g1*.5+g12*.25; !g1*2nd slope loading, g12*2nd slope loading2
p13 = p12+g2*.083+g22*.006889; !g2*1st slope loading, g22*1st slope loading2
p14 = p12+g2*.5+g22*.25; !g2*2nd slope loading, g22*2nd slope loading2
```

Appendix Q
Linear Model Equations and Mplus Syntax:
Gottfredson et al. (2014) Hybrid Model (k = 2) (Figure 15)

$$Y_{ti} = \lambda_1\eta_{0i} + \lambda_t\eta_{1i} + \lambda_{t^2}\eta_{2i} + \varepsilon_{ti} \quad (86)$$

$$\eta_{0i} = \alpha_{0k}^{p} + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (87)$$

$$\eta_{1i} = \alpha_{1k}^{p} + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (88)$$

$$\eta_{2i} = \alpha_{2k}^{p} \quad (89)$$

$$Logit(C_i|k = 2) = \pi_{00} \quad (90)$$

$$Summary_i = \gamma_{20k} + \gamma_{21}(RT_i) + \gamma_{22}(SFAS_i) \quad (91)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12 summary;

USEVARIABLES = RT - BINGE_12 s1-s3;

Missing = ALL (-99);

classes = c(2);

DEFINE: !Summary created here;
if (Pattern EQ 1) THEN summary = 4;
if (Pattern EQ 2) THEN summary = 3;
if (Pattern EQ 3) THEN summary = 3;
if (Pattern EQ 4) THEN summary = 2;
if (Pattern EQ 5) THEN summary = 3;
if (Pattern EQ 6) THEN summary = 2;
if (Pattern EQ 7) THEN summary = 2;
if (Pattern EQ 8) THEN summary = 1;

Analysis:
type = mixture;
ALGORITHM=INTEGRATION;
process = 8(starts);
starts = 400 100;

MODEL:
%OVERALL%
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
Q@0;
I S ON RT SFAS;
summary ON RT SFAS;
```

```
[ c#1* ] (logit_1); !Needed for pooling;

%C#1%
[ summary* ];
[ i* ] (i1_mean);
[ s* ] (s1_mean);
[ q* ] (q1_mean);

%C#2%
[ summary* ];
[ i* ] (i2_mean);
[ s* ] (s2_mean);
[ q* ] (q2_mean);

MODEL CONSTRAINT: !Pooling here;
 NEW(prob1);
 NEW(prob2);
 NEW(icept);
 NEW(slope);
 NEW(quad);

PROB1 = exp(logit_1)/(exp(0) + exp(logit_1));
PROB2 = 1 - prob1;

ICEPT = PROB1*I1_MEAN + PROB2*I2_MEAN;
SLOPE = PROB1*S1_MEAN + PROB2*S2_MEAN;
QUAD  = PROB1*Q1_MEAN + PROB2*Q2_MEAN;
```

Appendix R
Linear Model Equations and Mplus Syntax:
Selection Mixture, Model 1 (Figure 16)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (92)$$

$$\eta_{0i} = \alpha_{0k} + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (93)$$

$$\eta_{1i} = \alpha_{1k} + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (94)$$

$$\eta_{2i} = \alpha_2 \quad (95)$$

$$Logit(C_i|k) = \pi_{00} \quad (96)$$

$$Logit(s1_i) = \pi_{10} + \pi_{11}(BINGE_{ti})_{(t-1)} + \pi_{12}(BINGE_{ti}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (97)$$

$$Logit(s2_i) = \pi^*_{20} + \pi_{11}(BINGE_{ti})_{(t-1)} + \pi_{12}(BINGE_{ti}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (98)$$

$$Logit(s3_i) = \pi^*_{30} + \pi_{11}(BINGE_{ti})_{(t-1)} + \pi_{12}(BINGE_{ti}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (99)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 s1-s3;

Missing = ALL (-99);

categorical are s1-s3;

classes = c(2);

DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

ANALYSIS:
TYPE=MIXTURE;
algorithm = integration em;
integration = montecarlo (5000);
process = 8(starts);
starts = 400 100;

MODEL:
%OVERALL%
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
```

```
Q@0;
I S ON RT SFAS;

%C#1%
s1-s3 ON rt(1);
s1-s3 ON sfas(2);
s1-s3 ON binge_b(3);
s1-s3 ON binge_1(4);

[s1$1] (p2);
[s2$1] (p3);
[s3$1] (p4);

%C#2%
s1-s3 ON rt(1);
s1-s3 ON sfas(2);
s1-s3 ON binge_b(3);
s1-s3 ON binge_1(4);

[s1$1] (p2);
[s2$1] (p3);
[s3$1] (p4);

MODEL CONSTRAINT:
NEW(g1 g12);
p3 = p2+g1*.083+g12*.006889;
p4 = p2+g1*.5+g12*.25;
```

Appendix S
Linear Model Equations and Mplus Syntax:
Selection Mixture, Model 2 (Figure 16)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (100)$$

$$\eta_{0i} = \alpha_{0k} + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (101)$$

$$\eta_{1i} = \alpha_{1k} + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (102)$$

$$\eta_{2i} = \alpha_{2k} \quad (103)$$

$$Logit(C_i|k) = \pi_{00} \quad (104)$$

$$Logit(s1_i) = \pi_{10k} + \pi_{11_k}(BINGE_{ti})_{(t-1)} + \pi_{12_k}(BINGE_{ti}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (105)$$

$$Logit(s2_i) = \pi^*_{20k} + \pi_{11_k}(BINGE_{ti})_{(t-1)} + \pi_{12_k}(BINGE_{ti}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (106)$$

$$Logit(s3_i) = \pi^*_{30k} + \pi_{11_k}(BINGE_{ti})_{(t-1)} + \pi_{12_k}(BINGE_{ti}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (107)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 s1-s3;

Missing = ALL (-99);

categorical are s1-s3;

classes = c(2);

DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

ANALYSIS:
TYPE=MIXTURE;
algorithm = integration em;
integration = montecarlo (5000);
process = 8(starts);
starts = 400 100;

MODEL:
%OVERALL%
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
```

```
Q@0;
I S ON RT SFAS;

%C#1%
[s1$1] (p2);
[s2$1] (p3);
[s3$1] (p4);

s1-s3 ON RT(1);
s1-s3 ON SFAS(2);

s1 ON BINGE_B(3);
s2 ON BINGE_1(3);
s3 ON BINGE_6(3);
s1 ON BINGE_1(4);
s2 ON BINGE_6(4);
s3 ON BINGE_12(4);

%C#2%
[s1$1] (p12);
[s2$1] (p13);
[s3$1] (p14);

s1-s3 ON RT (1);
s1-s3 ON SFAS (2);

s1 ON BINGE_B(13);
s2 ON BINGE_1(13);
s3 ON BINGE_6(13);
s1 ON BINGE_1(14);
s2 ON BINGE_6(14);
s3 ON BINGE_12(14);

MODEL CONSTRAINT:
NEW(g1 g12 g2 g22);
p3 = p2+g1*.083+g12*.006889;
p4 = p2+g1*.5+g12*.25;
p13 = p12+g2*.083+g22*.006889;
p14 = p12+g2*.5+g22*.25;
```

Appendix T
Linear Model Equations and Mplus Syntax:
Selection Mixture, Model 3 (Figure 16)

$$Y_{ti} = \lambda_1 \eta_{0i} + \lambda_t \eta_{1i} + \lambda_{t^2} \eta_{2i} + \varepsilon_{ti} \quad (108)$$

$$\eta_{0i} = \alpha_{0k} + \gamma_{01}(RT_i) + \gamma_{02}(SFAS_i) + \zeta_{0i} \quad (109)$$

$$\eta_{1i} = \alpha_{1k} + \gamma_{11}(RT_i) + \gamma_{12}(SFAS_i) + \zeta_{1i} \quad (110)$$

$$\eta_{2i} = \alpha_{2k} + \gamma_{21}(RT_i) + \gamma_{22}(SFAS_i) + \zeta_{2i} \quad (111)$$

$$Logit(C_i|k) = \pi_{00} \quad (112)$$

$$Logit(s1_i) = \pi_{10k} + \pi_{11_{tk}}(BINGE_{ti})_{(t-1)} + \pi_{12_{tk}}(BINGE_{ti}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (113)$$

$$Logit(s2_i) = \pi^*_{20k} + \pi_{21_{tk}}(BINGE_{ti})_{(t-1)} + \pi_{22_{tk}}(BINGE_{ti}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (114)$$

$$Logit(s3_i) = \pi^*_{30k} + \pi_{31_{tk}}(BINGE_{ti})_{(t-1)} + \pi_{32_{tk}}(BINGE_{ti}) + \pi_{13}(RT_i) + \pi_{14}(SFAS_i) \quad (115)$$

```
DATA:
FILE IS BINGE.dat;

VARIABLE:
NAMES ARE ID RT SFAS BINGE_B BINGE_1 BINGE_6 BINGE_12;

USEVARIABLES = RT - BINGE_12 s1-s3;

Missing = ALL (-99);

categorical are s1-s3;

classes = c(2);

DATA MISSING:
names = BINGE_B BINGE_1-BINGE_12;
type = sdropout;
binary = S1-S3;

ANALYSIS:
TYPE=MIXTURE;
algorithm = integration em;
integration = montecarlo (5000);
process = 8(starts);
starts = 400 100;

MODEL:
%OVERALL%
I S Q | BINGE_B@0 BINGE_1@.083 BINGE_6@.5 BINGE_12@1;
```

```
Q@0;
I S ON RT SFAS;

%C#1%
[s1$1] (p2);
[s2$1] (p3);
[s3$1] (p4);

s1-s3 ON RT(1);
s1-s3 ON SFAS(2);
s1 ON BINGE_B BINGE_1;
s2 ON BINGE_1 BINGE_6;
s3 ON BINGE_6 BINGE_12;

%C#2%
[s1$1] (p12);
[s2$1] (p13);
[s3$1] (p14);

s1-s3 ON RT(1);
s1-s3 ON SFAS(2);
s1 ON BINGE_B BINGE_1;
s2 ON BINGE_1 BINGE_6;
s3 ON BINGE_6 BINGE_12;

MODEL CONSTRAINT:
NEW(g1 g12 g2 g22);
p3 = p2+g1*.083+g12*.006889;
p4 = p2+g1*.5+g12*.25;
p13 = p12+g2*.083+g22*.006889;
p14 = p12+g2*.5+g22*.25;
```